

☺ Effektsysteme ☺

Ingo Blechschmidt
<iblech@speicherleck.de>

5. November 2015
Augsburger Curry-Club

- 1 Spezifikation von Instruktionen
- 2 Freie Funktoren
- 3 Freie Monaden
- 4 Einschub: Koproduct von Monaden
- 5 Effektsysteme

Gewünschte Instruktionen

```
type St = ...  
data StateI :: * -> * where  
  Get :: StateI St  
  Put :: St -> StateI ()
```

Freie Funktoren und freie Monaden liefern ein allgemeines Konstruktionsrezept für Monaden mit gewünschter operationeller Semantik.

Gewünschte Instruktionen

```
type St = ...  
data StateI :: * -> * where  
  Get :: StateI St  
  Put :: St -> StateI ()
```

```
type Env = ...  
data ReaderI :: * -> * where  
  Ask :: ReaderI Env
```

```
type Log = ...  
data WriterI :: * -> * where  
  Tell :: Log -> WriterI ()
```

Gewünschte Instruktionen

```
type Err = ...
data ErrorI :: * -> * where
    Throw :: Err -> ErrorI a

data IOI :: * -> * where
    PutStrLn :: String -> IOI ()
    GetLine  :: IOI String
    Exit     :: IOI a
```

Freie Funktoren

Wir können aus einem Typkonstruktor $t :: * \rightarrow *$ auf unspektakulärste Art und Weise einen Funktor machen:

```
class Functor f where
    fmap :: (a -> b) -> (f a -> f b)

data FreeF t a = MkFreeF (exists r. (t r, r -> a))
data FreeF t a = forall r. MkFreeF (t r) (r -> a)
-- MkFreeF :: t r -> (r -> a) -> FreeF t a

liftF :: t a -> FreeF t a
liftF x = MkFreeF x id

instance Functor (FreeF t) where
    fmap phi (MkFreeF x h) = MkFreeF x (phi . h)
```


Wenn t ein Funktor *wäre*, so könnte man eine Funktion $r \rightarrow a$ zu einer Funktion $t\ r \rightarrow t\ a$ liften. Wenn t kein Funktor ist, geht das nicht.

Der freie Funktor $\text{FreeF } t$ über t *cheates*: Ein Wert vom Typ $\text{Free } t\ a$ besteht aus einem Wert $x :: t\ r$ zusammen mit einer Funktion $f :: r \rightarrow a$. Anschaulich stellen wir uns diese Kombination als das vor, was $\text{fmap } f\ x$ ergäbe, wenn t ein Funktor wäre.

Die Implementierung von $\text{fmap } \phi$ für $\text{FreeF } t$ ist dann rein *formal*: Anstatt ϕ wirklich auf einen Wert vom Typ $t\ r$ anzuwenden (was unmöglich ist), notieren wir uns nur die Information, dass ϕ anzuwenden *wäre*.


Das Paar bestehend aus dem *freien Funktor* $\text{FreeF } t$ über t und der Funktion liftF erfüllt folgende *universelle Eigenschaft*:

Ist G irgendein Funktor und $\text{phi} :: t \text{ a} \rightarrow G \text{ a}$ irgendeine Funktion, so existiert genau eine Funktion $\text{phi}' :: \text{FreeF } t \text{ a} \rightarrow G \text{ a}$ mit $\text{phi}' \circ \text{liftF} = \text{phi}$.



Einen Typkonstruktor $t :: * \rightarrow *$ kann man auch als einen Funktor $\text{Hask}_0 \rightarrow \text{Hask}$ auffassen. Dabei hat die Kategorie Hask_0 dieselben Objekte wie Hask , enthält aber nur Identitätsmorphisimen.

Der freie Funktor über t ist dann die *Links-Kan-Erweiterung* von t längs der Inklusion $\text{Hask}_0 \hookrightarrow \text{Hask}$. Die angegebene Definition von $\text{FreeF } t$ ist nichts anderes als die *Koendeformel* für Links-Kan-Erweiterungen.



$$\text{FreeF } t \ a = \int^{r \in \text{Hask}_0} (t \ r, r \rightarrow a)$$

**CAN WE EXTEND
THIS FUNCTOR?**

$$\begin{array}{ccc} \mathcal{C} & & \\ \uparrow p & \dashrightarrow \text{Lan}_p F & \\ \mathcal{M} & \xrightarrow{F} & \mathcal{A} \end{array}$$

YES WE KAN

Beispiel: Zustand

```
class Functor f where
  fmap :: (a -> b) -> (f a -> f b)

data FreeF t a = forall r. MkFreeF (t r) (r -> a)

data StateI :: * -> * where
  Get :: StateI St
  Put :: St -> StateI ()

-- 'FreeF StateI' ist isomorph zu:
data StateF a = Get (St -> a) | Put St a
```

Offensichtlich wird `StateF` zu einem Funktor. Das zugehörige `fmap` ist recht langweilig:

```
instance Functor StateF where
  fmap phi (Get k)      = Get (f . k)
  fmap phi (Put st k) = Put st (phi k)
```

Genauso langweilig sind die Funktor-Instanzen von anderen Typkonstruktoren der Form `FreeF t`.

Freie Monaden

Wir können aus einem Funktor $f :: * \rightarrow *$ auf unspektakulärste Art und Weise eine Monade machen:

```
data FreeM f a = Pure a | Roll (f (FreeM f a))
```

```
instance (Functor f) => Monad (FreeM f) where  
  return x = Pure x
```

```
Pure x >>= k = k x
```

```
Roll u >>= k = Roll (fmap (>>= k) u)
```

Mehr zu freien Monaden in einem separaten Foliensatz:

<http://curry-club-augsburg.de/posts/>

[2015-08-14-ankuendigung-siebtes-treffen.html](http://curry-club-augsburg.de/posts/2015-08-14-ankuendigung-siebtes-treffen.html)

Zusammengesetzt

```
data FreeF t a = forall r. MkFreeF (t r) (r -> a)
```

```
data FreeM f a = Pure a | Roll (f (FreeM f a))
```

```
-- Also ist 'FreeM (FreeF t) a' isomorph zu:
```

```
data Prog t a =
```

```
  Pure a | forall r. Step (t r) (r -> Prog t a)
```

```
data StateI :: * -> * where
```

```
  Get :: StateI St
```

```
  Put :: St -> StateI ()
```

```
-- 'Prog StateI a' ist isomorph zu:
```

```
data StateProg a
```

```
  = Pure a
```

```
  | Get (St -> StateProg a)
```

```
  | Put St (StateProg a)
```

Auch ohne die Motivation über freie Funktoren und freie Monaden besitzt $\text{Prog } t \ a$ eine anschauliche Bedeutung. Wir können uns einen Wert vom Typ $\text{Prog } t \ a$ als eine Folge von Instruktionen vorstellen, die schlussendlich einen Wert vom Typ a produziert. Welche Instruktionen vorkommen können, entscheidet t .

Ein Wert vom Typ $\text{Prog } t \ a$ ist entweder von der Form $\text{Pure } x$, also ein triviale Folge von Instruktionen mit Produktionswert x , oder von der Form $\text{Step } u \ k$. Dabei kodiert u eine Instruktion, die bei Ausführung einen Wert $x :: r$ produziert, und die Continuation $k :: r \rightarrow \text{Prog } t \ a$ gibt an, wie es danach weitergehen soll.

Operationelle Semantik

```
data StateProg a
```

```
  = Pure a
```

```
  | Get    (St -> StateProg a)
```

```
  | Put St (StateProg a)
```

```
interpret :: StateProg a -> (St -> (a, St))
```

```
interpret (Pure x)      st = (x, st)
```

```
interpret (Get k)       st = interpret (k st) st
```

```
interpret (Put st' u)   st = interpret u      st'
```

In Kürze

-- Die freie Monade über dem freien Funktor über t :

```
data Prog t a =  
  Pure a | forall r. Step (t r) (r -> Prog t a)
```

- Werte vom Typ `Prog t a` sind rein syntaktische Beschreibungen von Aktionsfolgen. Insbesondere gelten keinerlei besondere Rechenregeln, wie zum Beispiel `Put x (Put x' m) == Put x' m`.
- Erst durch Angabe eines Interpreters wird die Konstruktion zum Leben erweckt.
- Man kann leicht Instruktionsspezifikationen miteinander kombinieren!
- In der naiven Implementierung: Effizienzproblem mit linksassoziativer Verwendung von `(>>=)`

Ein und dieselbe freie Monade kann mehrere verschiedene Interpreter zulassen. Zum Beispiel kann man in Produktion einen Interpreter verwenden, der eine Datenbank anspricht, und zum Testen einen, der fiktive Werte vortäuscht (Mocking).

Siehe unbedingt auch:

- Heinrich Apfelmus. *The Operational Monad Tutorial*. 2010.
<http://apfelmus.nfshost.com/articles/operational-monad.html>

Beispiel: Einfaches Multitasking

```
data ProcessI :: (* -> *) -> * -> * where
  Lift  :: m a -> ProcessI m a
  Stop  :: ProcessI m a
  Fork  :: ProcessI m Bool  -- wie in Unix
  Yield :: ProcessI m ()

-- Interpreter, der nur bei Aufruf von 'Yield' die
-- Kontrolle an den nächsten Prozess weitergibt
runCoop  :: (Monad m) => Prog (ProcessI m) a -> m ()

-- Interpreter, der nach jeder Aktion in der Basis-
-- monade die Kontrolle weitergibt
runForced :: (Monad m) => Prog (ProcessI m) a -> m ()
```

Siehe Beispielcode:

[https://github.com/iblech/vortrag-haskell/blob/master/
effektsysteme.hs](https://github.com/iblech/vortrag-haskell/blob/master/effektsysteme.hs)

Einschub: Koproduct von Monaden

Sind m und n Monaden, so kann man eine Monade bauen, die die Fähigkeiten von m und n vereint. Diese heißt *Koproduct* von m und n .

```
data Sum      m n a = Inl (m a) | Inr (n a)
type Coprod  m n a = Prog (Sum m n) a

-- Coprod m n vereint m und n:
inl :: m a -> Coprod m n a      inr :: n a -> Coprod m n a
inl x = Step (Inl x) Pure      inr x = Step (Inr x) Pure

-- Ausführung mit (universelle Eigenschaft):
elim :: (Monad m, Monad n, Monad s)
      => (m a -> s a) -> (n a -> s a)
      -> (Coprod m n a -> s a)
elim = ...
```

Beispiel: State s \sqcup Error e

```
type Err e = Either e
```

```
ex :: Coprod (State Int) (Err String) Int
```

```
ex = do
```

```
  st <- inl get
```

```
  if st <= 0 then inr (Left "Fehler") else do
```

```
    inl $ put (st - 1)
```

```
  return $ st^2 + st + 1
```

Ausführung durch Angabe einer Monade, in die man State Int und Err String einbetten kann – zum Beispiel IO:

```
runM :: (Show e) => s -> Coprod (State s) (Err e) a -> IO a
runM st m = do
```

```
  ref <- newIORef st
  elim (embedState ref) embedErr m
```

```
embedState :: IORef s -> State s a -> IO a
```

```
embedState ref m = do
  st <- readIORef ref
  let (x,st') = runState m st
  writeIORef ref st'
  return x
```

```
embedErr :: (Show e) => Err e a -> IO a
```

```
embedErr (Left e) = putStrLn ("Fehler: " ++ show e) >> exit 1
embedErr (Right x) = return x
```

Effektsysteme

Effektsysteme lösen das Problem der fehlenden Kompositionalität von Monaden.

Die Monade `Eff r` ist wie `Prog t`, nur

- performant bezüglich (`>>=`) und
- mit `r` als *Liste* von möglichen Instruktionen (auf Typebene) anstatt mit `t` als Instruktionen kodierenden *Typkonstruktor*.

```
ask :: (Member (Reader env) r) => Eff r env
get :: (Member (State st) r) => Eff r st
put :: (Member (State st) r) => st -> Eff r ()
-- Typen schreiben r nicht eindeutig vor!
```

Wenn man nachträglich seinen Transformerstack ändert, muss man viele Typsignaturen und gelegentlich auch einige Codefragmente anpassen (zum Beispiel `lift` in `lift . lift` ändern). Das ist mühsam.

Bei Verwendung von `Eff` muss man das nicht.

Werte vom Typ `Eff r a` sind wie bei `Prog t a` auch nur *Beschreibungen* von auszuführenden Aktionen; erst durch Angabe von Interpretern können sie ausgeführt werden. Interpreter können leicht miteinander kombiniert werden.

Interpreter

```
type Env = ...
```

```
data Reader :: * -> * where
```

```
  Get :: Reader Env
```

```
ask :: (Member Reader r) => Eff r Env
```

```
ask = Roll (inj Get) (tsingleton Pure)
```

```
runReader :: Env -> Eff (Reader ': r) a -> Eff r a
```

```
runReader e m = loop m where
```

```
  loop (Pure x)    = return x
```

```
  loop (Roll u q) = case decomp u of
```

```
    Right Get -> loop $ qApp q e
```

```
    Left  u    -> Roll u (tsingleton (qComp q loop))
```

```
-- kürzer:
```

```
runReader e = handleRelay return (\Get k -> k e)
```

Siehe unbedingt:

- Oleg Kiselyov, Hiromi Ishii. *Freer Monads, More Extensible Effects*. 2015.

<http://okmij.org/ftp/Haskell/extensible/more.pdf>

- Oleg Kiselyov, Amr Sabry, Cameron Swords. *Extensible Effects. An Alternative to Monad Transformers*. 2013.

<http://okmij.org/ftp/Haskell/extensible/exteff.pdf>

- Andrej Bauer, Matija Pretnar. *Programming with Algebraic Effects and Handlers*. 2012.

<http://arxiv.org/abs/1203.1539>